Project 1: Goblet Design

Douglas Meade, Ronda Sanders, and Xian Wu
Department of Mathematics

Instructions

The goal of the project is to design the most visually appealing goblet that meets the following criteria:

- the goblet will be molded using a symmetric mold, that is, the goblet must be a solid of revolution;
- the goblet must hold between 237 and 266 cm^3 (8–9 oz) of liquid and uses less than 200 cm^3 of glass;
- the height of the center of mass must be less than 3 times the radius of the foot so it would be reasonably stable;
- thickness of the glass must be at least 0.25 cm at its thinnest point.

Your report is due at the beginning of the next lab meeting and should follow the guidelines set forth in the What is a Project Report? on our lab web. In particular, your report should include the following:

- a detailed description of your design.
- a (2-D) plot of the region to be revolved and a (3-D) plot of the goblet
- detailed numerical results showing that the criteria are satisfied.

An Example in Maple

Most maple commands used here were introduced in past labs. In particular, you should review last semester's Lab H: (http://www.math.sc.edu/calclab/141L-F05/labs/MathModelRollerCoaster05f.pdf) Mathematical Models-Design a Roller Coaster and, of course, the last lab.

1. Basic Analyses: By studying you favorite goblets at home, you decide that your new goblet should have a foot (base) of 1.7 inches in radius and 0.5 inches in height. You also like to have a twisted stem of 6 inches long and 0.1-0.2 inches in radius. Finally, you want it to have a bowl that gradually extends to a radius of 1 inch in the first 1.5 inches from the stem and then to a radius of 2 inches in the last 1.5 inches. Put it in sideways and your goblet hence extends from $x = 0$ to $x = 9.5$. The outside profile of your goblet can therefore be described by a piecewise function that consists of four functions defined over $0 \leq x < 0.5$, $0.5 \leq x < 6.5$, $6.5 \leq x < 8$, and $8 \leq x \leq 9.5$, respectively.
2. Working with Maple
   (a) You should always start a Maple session with `restart` command as it clears the internal memory so that Maple acts (almost) as if just started and is very helpful in case that you made a mistake and want to start over.
   ```
   > restart;
   ```
   (b) Load needed packages:
   ```
   > with(plots):
   > with(Student[Calculus1]):
   ```
   (c) For the foot, it is easy to see that we should define $f1(x) = 1.7 - 3x$.
   ```
   > f1:=x->1.7-3*x;
   ```

(d) For a plain stem of radius 0.1, we may just set $f2(x) = 0.1$. We will add some small waves to it with $a|sin(b(x-c))|$. For this particular goblet, we take $a = 0.1$ as the height of waves, $b = \pi$ to have six waves for our stem from $x = 0.5$ to $x = 6.5$, and $c = 0.5$ so that waves start at $x = 0.5$.
```
> f2:=x->0.1+0.1*abs(sin(Pi*(x-0.5)));
```
(e) The first part of the bowl starts at $(6.5, 0.1)$ and should be concave down. So let's use $f3(x) = a\sqrt{x-6.5} + 0.1$, where a is a constant to be solved late to satisfy the condition at $x = 8$.
```
> f3:=x->a*sqrt(x-6.5)+0.1;
> eq1:=f3(8)=1;
> df3:=diff(f3(x),x);
```
(f) The second part of the bowl should be concave up and let us try $f4(x) = bx^2 + cx + d$.
```
> f4:=x->b*x^2+c*x+d;
> df4:=diff(f4(x),x);
```
(g) Of course, we need to set up three more equations so that $f4(x)$ will be smoothly connected to $f3(x)$ at $x = 8$ and satisfy the condition at the end $x = 9.5$.
```
> eq2:=f4(8)=f3(8);
> eq3:=eval(df4,x=8)=eval(df3,x=8);
> eq4:=f4(9.5)=2;
```
(h) Solve for a, b, c, d using the `solve` command and assign the solutions to values.
```
> values:=solve({eq1,eq2,eq3,eq4},{a,b,c,d});
```
(i) Plug those solved values into functions once for all.
```
> assign(values);
```
(j) Put all four functions together and plot it to see the outside profile.
```
> F:=x->piecewise(x<=0.5,f1(x),x>0.5 and x<=6.5,f2(x),x>6.5 and x<=8,f3(x),
8<x,f4(x));
> plot(F(x),x=0..9.5,y=0..5,scaling=constrained);
```
(k) Shift outside of the bowl 0.2 inches to the right to get the inside profile.
```
> G:=x->piecewise(x<=6.7,0,x>6.7 and x<=8.2,f3(x-0.2), 8.2<x,f4(x-0.2));
> plot(G(x),x=0..9.5,y=0..5,scaling=constrained);
```
(l) Here is the goblet and let's check the criteria (notice 1 inch=2.54 cm). It is easy to see that the center of mass is located on the x-axis, say, at $x = CH$. A formula for CH is given below and it can be obtained in a similar way for the formula of the volume of a solid of revolution.
```
> VolumeOfRevolution(F(x),G(x),x=0..9.5,output=plot,orientation=[0,180],
title=‘‘ ’’);
> Capacity:=evalf(2.54^3*VolumeOfRevolution(G(x),x=0..9.5,output=integral));
> Glass:=evalf(2.54^3*VolumeOfRevolution(F(x),G(x),x=0..9.5,output=integral));
> Difference:=x->F(x)-G(x);
> plot([2.54*Difference(x),1/4],x=6.7..9.5);
> Thickness:=2.54*Difference(8.05);
> CH:=evalf(int(x*((F(x))^2-(G(x))^2),x=0..9.5)/int((F(x))^2-(G(x))^2,x=0..9.5));
> Ratio:=CH/F(0);
```

3. Conclusion: Looks like we have a stable goblet that satisfies the requirement on amount of glass used. However, the goblet does not hold enough and the glass is too thin.

Acknowledgement

- This project is based on a project created in the Department of Mathematics at Kenyon College.
- The best design from each section and an over-all winner will be selected. We want to thank MapleSoft for providing prizes for winning projects.